Definition. *For any functions* $M, N: \mathbb{R}^2 \to \mathbb{R}$, $M(x,y)\mathrm{d}x + N(x,y)\mathrm{d}y$ *is a differential form.*

Definition. *The differential form* $M(x,y)\mathrm{d}x + N(x,y)\mathrm{d}y$ *is said to be exact if* $\exists u \in C^1(\mathbb{R}^2)$ *such that*

$$\mathrm{d}u = M(x,y)\mathrm{d}x + N(x,y)\mathrm{d}y \Rightarrow \frac{\partial u}{\partial x} = M(x,y), \frac{\partial u}{\partial y} = N(x,y)$$

Definition. $M(x,y)\mathrm{d}x + N(x,y)\mathrm{d}y = 0$ *is called a differential equation.*

- If $M(x,y)\mathrm{d}x + N(x,y)\mathrm{d}y$ is exact, the associated differential equation has the solution

  $$u(x,y) = c,$$

  an implicit solution for the function $y(x)$.
- For an exact differential form with $u \in C^2(\mathbb{R}^2)$ we have $\frac{\partial M}{\partial y} = \frac{\partial^2 u}{\partial y \partial x} = \frac{\partial^2 u}{\partial x \partial y} = \frac{\partial N}{\partial x} \Rightarrow \frac{\partial M}{\partial y} = \frac{\partial N}{\partial x}$

Example. $\cos(x+y)\,\mathrm{d}x + (3y^2 + 2y + \cos(x+y))\,\mathrm{d}y = 0.$

```
In[33]:= Dform = Cos[x+y] dx + (3y^2+2y+Cos[x+y]) dy; Deq = Dform == 0
```

$\mathrm{dx}\cos(x+y) + \mathrm{dy}(\cos(x+y) + 3y^2 + 2y) = 0$

```
In[34]:= {Mf[x_,y_],Nf[x_,y_]} = {Coefficient[Dform,dx],Coefficient[Dform,dy]}
```

$\{\cos(x+y), \cos(x+y) + 3y^2 + 2y\}$

```
In[35]:= D[Mf[x,y],y] == D[Nf[x,y],x]
```

True

- Assume x, y are independent variables and integrate

$$\frac{\partial u}{\partial x} = M(x,y)$$
$$\mathrm{d}u = M(x,y)\mathrm{d}x$$
$$\int \mathrm{d}u = \int M(x,y)\mathrm{d}x$$
$$u = \int M(x,y)\mathrm{d}x + k(y)$$
$$\frac{\partial u}{\partial y} = N(x,y) = \int \frac{\partial M(x,y)}{\partial y}\mathrm{d}x + k'(y)$$
$$k'(y) = N(x,y) - \int \frac{\partial M(x,y)}{\partial y}\mathrm{d}x$$

$$\frac{\partial u}{\partial y} = N(x,y)$$
$$\mathrm{d}u = N(x,y)\mathrm{d}y$$
$$\int \mathrm{d}u = \int N(x,y)\mathrm{d}y$$
$$u = \int N(x,y)\mathrm{d}y + l(x)$$
$$\frac{\partial u}{\partial x} = M(x,y) = \int \frac{\partial N(x,y)}{\partial x}\mathrm{d}y + l'(x)$$
$$l'(x) = M(x,y) - \int \frac{\partial N(x,y)}{\partial x}\mathrm{d}y$$

Example. (continued) $\cos(x+y)\,\mathrm{d}x + (3y^2 + 2y + \cos(x+y))\,\mathrm{d}y = 0.$

```
In[36]:= leq = l'[x] == Mf[x,y] - Integrate[D[Nf[x,y],x],y]
```

$l'(x) = \sin(x)\sin(y) - \cos(x)\cos(y) + \cos(x+y)$

```
In[37]:= sol = DSolve[leq,l[x],x][[1,1]]
```

$l(x) \to c_1$

```
In[38]:= u[x_,y_] = Integrate[Nf[x,y],y] + l[x] /. sol
```

$c_1 + \sin(x)\cos(y) + \cos(x)\sin(y) + y^3 + y^2$

```
In[39]:=
```

- Verify

$$\frac{\partial u}{\partial x} = M(x,y) \, \Big| \, \frac{\partial u}{\partial y} = N(x,y)$$

Example. (continued) $\cos(x+y)\,\mathrm{d}x + (3y^2+2y+\cos(x+y))\,\mathrm{d}y = 0.$

```
In[39]:= {u[x,y],Mf[x,y],Nf[x,y]}
```

$\{c_1 + \sin(x)\cos(y) + \cos(x)\sin(y) + y^3 + y^2, \cos(x+y), \cos(x+y) + 3\,y^2 + 2\,y\}$

```
In[40]:= Simplify[{D[u[x,y],x] == Mf[x,y],D[u[x,y],y] == Nf[x,y]}]
```

{True, True}

- Consider the DE $-y\,\mathrm{d}x + x\,\mathrm{d}y = 0$. Since

$$\frac{\partial}{\partial y}(-y) = -1 \neq 1 = \frac{\partial}{\partial x}(x),$$

  the DE is not exact
- It may be possible to multiply the DE $P(x,y)\,\mathrm{d}x + Q(x,y)\,\mathrm{d}y = 0$ with another function $F(x,y) \neq 0$

$$M(x,y)\,\mathrm{d}x + N(x,y)\,\mathrm{d}y = 0,$$
$$M(x,y) = F(x,y)P(x,y), N(x,y) = F(x,y)Q(x,y),$$

  and obtain and exact DE with $\partial_x N = \partial_y M$

Example. $F = 1/x^2$, $-y\,\mathrm{d}x + x\,\mathrm{d}y = 0$.

```
In[42]:= df1 = -y dx + x dy; {P[x_,y_],Q[x_,y_]}={Coefficient[df1,dx],Coefficient[df1,dy]}
```

$\{-y, x\}$

```
In[43]:= D[P[x,y],y] == D[Q[x,y],x]
```

False

```
In[44]:= F[x,y]=1/x^2; D[F[x,y] P[x,y],y] == D[F[x,y] Q[x,y],x]
```

True

```
In[45]:=
```

Theorem 1. *If the coefficients* $P(x,y), Q(x,y)$ *of the differential form* $P(x,y)\mathrm{d}x + Q(x,y)\mathrm{d}y$ *satisfy*

$$R(x) = \frac{1}{Q(x,y)}\left(\frac{\partial P}{\partial y} - \frac{\partial Q}{\partial x}\right),$$

then

$$F(x) = \exp\int R(x)\,\mathrm{d}x,$$

is an integrating factor of the differential form.

Theorem 2. *If the coefficients* $P(x,y), Q(x,y)$ *of the differential form* $P(x,y)\mathrm{d}x + Q(x,y)\mathrm{d}y$ *satisfy*

$$R^*(y) = \frac{1}{P(x,y)}\left(\frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y}\right),$$

then

$$F^*(y) = \exp\int R^*(y)\,\mathrm{d}y,$$

is an integrating factor of the differential form.

Example. $P(x,y)\,\mathrm{d}x+Q(x,y)\mathrm{d}y=(e^{x+y}+ye^y)\,\mathrm{d}x+(x\,e^y-1)\,\mathrm{d}y=0,\; y(0)=1$

```
In[1]:= P[x_,y_]=Exp[x+y]+ y Exp[y]; Q[x_,y_]=x Exp[y]-1; IniCond = y[0]==1
```

$y(0)=1$

```
In[4]:= IsExact = Simplify[D[P[x,y],y] - D[Q[x,y],x] == 0]
```

$e^y\,(e^x+y)=0$

```
In[5]:= R = Simplify[ (D[P[x,y],y] - D[Q[x,y],x])/Q[x,y] ]
```

$$\frac{e^y\,(e^x+y)}{x\,e^y-1}$$

```
In[6]:= Rs = Simplify[ (D[Q[x,y],x] - D[P[x,y],y])/P[x,y] ]
```

-1

```
In[7]:= Fs[y_] = Exp[Integrate[Rs,y]]
```

e^{-y}

```
In[9]:= IsExact = Simplify[D[Fs[y] P[x,y],y] - D[Fs[y] Q[x,y],x] == 0]
```

True

```
In[10]:=
```